/*
 * This file is part of the Try! Engine project.
 *
 * Copyright 2010-2011 Emanuele Bertoldi. All rights reserved.
 *
 * THIS SOFTWARE IS PROVIDED BY THE COPYRIGHT HOLDERS AND CONTRIBUTORS "AS IS"
 * AND ANY EXPRESS OR IMPLIED WARRANTIES, INCLUDING, BUT NOT LIMITED TO, THE
 * IMPLIED WARRANTIES OF MERCHANTABILITY AND FITNESS FOR A PARTICULAR PURPOSE
 * ARE DISCLAIMED. IN NO EVENT SHALL THE COPYRIGHT HOLDER OR CONTRIBUTORS BE
 * LIABLE FOR ANY DIRECT, INDIRECT, INCIDENTAL, SPECIAL, EXEMPLARY, OR
 * CONSEQUENTIAL DAMAGES (INCLUDING, BUT NOT LIMITED TO, PROCUREMENT OF
 * SUBSTITUTE GOODS OR SERVICES; LOSS OF USE, DATA, OR PROFITS; OR BUSINESS
 * INTERRUPTION) HOWEVER CAUSED AND ON ANY THEORY OF LIABILITY, WHETHER IN
 * CONTRACT, STRICT LIABILITY, OR TORT (INCLUDING NEGLIGENCE OR OTHERWISE)
 * ARISING IN ANY WAY OUT OF THE USE OF THIS SOFTWARE, EVEN IF ADVISED OF THE
 * POSSIBILITY OF SUCH DAMAGE.
 *
 * You should have received a copy of the modified BSD License along with this
 * program. If not, see <http://www.opensource.org/licenses/bsd-license.php>
 */

import QtQuick 1.0

Item {
    id: main

    enabled: widget.enabled

    SystemPalette {
        id: palette

        colorGroup: SystemPalette.Active
    }

    Rectangle {
        anchors.fill: parent
        anchors.topMargin: 9
        anchors.bottomMargin: 9
        anchors.leftMargin: 9
        anchors.rightMargin: 9
        border.width: 1
        border.color: palette.dark

        GridView {
            id: grid

            property int firstIndexDrag: -1

            cellWidth: 72
            cellHeight: 72
            model: assets
            delegate: assetDelegate
            anchors.fill: parent
            anchors.topMargin: 5
            anchors.bottomMargin: 5
            anchors.leftMargin: 5
            anchors.rightMargin: 5
            visible: (grid.count > 0) ? true : false
            clip: true

            Behavior on opacity { NumberAnimation { duration: 250 } }

            onFirstIndexDragChanged: {

                var dragging = (grid.firstIndexDrag != -1) ? true : false;
                
                grid.opacity = (dragging) ? 0.2 : 1.0;
                grid.interactive = (dragging) ? false : true;

                if (dragging)
                    widget.startDraggingAsset(grid.firstIndexDrag);
            }

            Connections {
                 target: widget
                 onAssetDropped: { grid.firstIndexDrag = -1; }
            }

            MouseArea {
                id: coords

                anchors.fill: parent

                onReleased: { grid.firstIndexDrag = -1; }
                onPressed: { grid.firstIndexDrag = grid.indexAt(mouseX, mouseY); }
            }

            Component {
                id: assetDelegate

                Rectangle {
                    id: asset

                    width: grid.cellWidth
                    height: grid.cellHeight
                    color: "transparent"
                    opacity: (asset.enabled) ? 1.0 : 0.2

                    Behavior on opacity { NumberAnimation { duration: 250 } }

                    Column {
                        anchors.fill: parent
                        anchors.topMargin: 10
                        anchors.bottomMargin: 10
                        spacing: 5

                        Image {
                            source: (icon) ? icon : "../icons/asset.png"
                            width: 32
                            fillMode: Image.PreserveAspectFit
                            anchors.horizontalCenter: parent.horizontalCenter
                        }

                        Text {
                            text: name
                            width: parent.width
                            horizontalAlignment: Text.AlignHCenter
                            elide: Text.ElideRight
                        }
                    }

                    states: [
                        State {
                            name: "Dragging"; when: index == grid.firstIndexDrag
                            PropertyChanges { target: asset; color: palette.highlight }
                        }
                    ]
                }
            }
        }
    }

    Text {
        anchors.centerIn: parent
        text: qsTr("No assets.")
        visible: (grid.count > 0) ? false : true
        color: palette.dark
    }
}
